
Règles de la déduction naturelle
Liens entre les tactiques Coq et la déduction naturelle

3 Logique intuitionniste vs classique

4 Curryfication

5 Sémantique de Heyting-Kolmogorov

6 Correspondance de Curry-Howard

7 Structures de données inductives
Sans récursion

Objectifs du cours

1 Mieux comprendre ce qu’est une preuve, apprendre à raisonner
rigoureusement.

2 Découvrir la preuve de programmes et de théorèmes.

3 Aspects pratiques: Coq

4 Aspects théorique: correspondance de Curry-Howard, . . .

Comment combattre le pire ennemi de l’informaticien: le bug

Logiciels critiques

Nucléaire

Transports

Santé

. . .

Exemple ”Ariane Vol 501”
”Because of the different flight path, a data conversion from a
64-bit floating point to 16-bit signed integer value caused a
hardware exception (more specifically, an arithmetic overflow, as
the floating point number had a value too large to be
represented by a 16-bit signed integer). Efficiency considerations
had led to the disabling of the software handler (in Ada code) for
this error trap, although other conversions of comparable
variables in the code remained protected. This caused a cascade
of problems, culminating in destruction of the entire flight.”

Exemple ”Mars Climate Orbiter”

”The ’root cause’ of the loss of the spacecraft was the failed
translation of English units into metric units in a segment of
ground-based, navigation-related mission software.”

Coût : $327 600 000.

Un logiciel pour corriger les bugs ?

Problème de l’arrêt

Il n’existe pas de programme permettant de décider si un programme
termine ou pas.

Qu’est-ce qu’un bug ?

C’est un programme qui ne respecte pas ses spécifications.

Première idée

Tester !
On écrit une fonction:

Programme + Propriété → Vrai/Faux

Mais çà ne suffit pas !

Interprétation abstraite

On réduit la classe des propriétés que l’on va vérifier.

Exemple

2003 Astrée vérifie l’absence d’erreur d’execution (RTE) dans le
système de controle de vol de l’Airbus A340, (132,000 lignes
de C).

2004 Astrée vérifie l’absence d’erreur d’execution (RTE) dans des
systèmes pour l’Airbus A380.

2008 Astrée vérifie l’absence d’erreur d’execution (RTE) dans le
système d’accrochage du véhicule automatique de transfert
européen (ATV).

Narboux et Magaud (Unistra) Preuves Assistées par Ordinateur 2018/2019 14 / 177



Problème

On a une infinité de cas.

Solution

Réaliser une preuve.
Un raisonnement fini pour traiter une infinité de cas.

Qu’est-ce qu’une preuve ?

un argument convainquant

une suite de déductions à partir des axiomes

un algorithme (correspondance de Curry-Howard)

Oui mais...

Il peut être difficile de se convaincre qu’une preuve est correcte :

le nombre de théorèmes

présence de calculs

trop de détails techniques, trop de cas

la taille de la preuve

La problème de la vérification d’une preuve
Présence de calculs

Théorème des 4 couleurs

Quatre couleurs suffisent pour colorier une carte géographique plane sans
que deux pays ayant une frontière en commun ne soient de la même
couleur.

1976 Appel and Hake (1478 configurations, 1200 heures de calcul)

2004 Formalisation en Coq par Gonthier et Werner

Le problème de la vérification d’une preuve
Présence de calculs

Conjecture de Kepler/Théorème de Hales

Pour un empilement de sphères égales, la densité maximale
est atteinte pour un empilement cubique à faces centrées.

1998 Preuve par Thomas Hales

2004 - 2014 Projet Flyspeck: formalisation du théorème en
cours en HOL-light avec des contributions en
Coq et Isabelle (plus de 300000 lignes) Photo par Robert

Cudmore

Robert MacPherson, éditeur, écrit que:

“The news from the referees is bad, from my perspective. They have not been able
to certify the correctness of the proof, and will not be able to certify it in the
future, because they have run out of energy to devote to the problem. This is not
what I had hoped for. The referees put a level of energy into this that is, in my
experience, unprecedented. ”

Le problème de la vérification d’une preuve I
La taille de la preuve

Théorème de Feit-Thompson

Theorem Feit_Thompson (gT:finGroupType) (G:{group gT}):

odd ##|G| -> solvable G.

Preuve en Coq par Georges Gonthier et son équipe (septembre 2012)a:
170 000 lignes, 15 000 définitions, 4 200 théorèmes

ahttp://ssr2.msr-inria.inria.fr/~jenkins/current/progress.html

Le problème de la vérification d’une preuve II
La taille de la preuve

Théorème de Fermat-Wiles

Il n’existe pas de nombres entiers non nuls , et tels que :

xn + yn = zn

dès que est un entier strictement supérieur à 2.

Paul Wolfskehl offre 100,000 marks

deux conditions: relues par les pairs, attendre 2 ans

1907-1908 : 621 tentatives

La problème de la vérification d’une preuve
Quelques succès

Un compilateur (CompCert: compilateur C prouvé en Coq, Xavier
Leroy et. al.)

Un système d’exploitation (seL4: micro kernel prouvé en Isabelle,
Gerwin Klein et. al.)

Un système de paiement (Gemalto, Andronick et. al.)

Système de pilotage d’une ligne de métro (ligne D à Lyon, ligne 14 à
Paris)

Une fonction de hash (SHA 256, Andrew Appel, 2015)

Un protocole cryptographique (OpenSSL HMAC, Andrew Appel et.
al., 2015)

Une quête de la rigueur

1 Clarifier les hypothèses

2 Clarifier ce qu’est une preuve

3 Être si précis que l’on a plus besoin de comprendre la preuve pour la
vérifier

4 Automatiser des preuves

Par définition vérifier qu’une preuve est correcte est un problème décidable.

On peut donc construire des assistants de preuve.

Exemples

Coq

Isabelle

PVS

HOL

Qu’est-ce que Coq ?

un assistant de preuve

développé et distribué librement par l’INRIA

Il permet de :

définir des notions mathématiques et/ou des programmes

démontrer mécaniquement des théorèmes mathématiques mettant en
jeu ces définitions

ACM Software System Award

2015 GCC

2014 Mach

2013 Coq

2012 LLVM

2011 Eclipse

2010 GroupLensCFRS

2009 VMware

2008 Gamma Parallel Database System

2007 Statemate

2006 Eiffel

2005 The Boyer-Moore Theorem Prover

2004 Secure Network Programming

2003 Make

2002 Java

. . .

1991 TCP-IP

Exemples d’énoncés mathématiques

Le théorème de Pythagore.

”Si les portes sont ouvertes c’est que la rame est en face d’un quai”.

Processus de preuve

Les deux étapes du développement d’une démonstration dans Coq sont les
suivantes :

d’abord la construction interactive d’une démonstration par
l’utilisateur ;

ensuite la vérification automatique de la correction de démonstration
par le système.

L’utilisateur prouve, puis le système vérifie que la preuve est bien
correcte.

Ressources disponibles

Site de l’INRIA consacré à Coq:
I Téléchargement du logiciel:

http://coq.inria.fr/distrib-fra.html

I Manuel de référence Coq:

http://coq.inria.fr/doc/

Livre et recueil d’exercices sur Coq:
I Coq’Art par Y. Bertot et P. Castéran (en français en ligne, en anglais

ou en chinois à la bibliothèque)

http://www.labri.fr/perso/casteran/CoqArt/coqartF.pdf

I Software Foundations par Benjamin C. Pierce, Chris Casinghino,
Michael Greenberg,Vilhelm Sjöberg, Brent Yorgey

http://www.cis.upenn.edu/~bcpierce/sf/

I D’autres cours en ligne sont disponibles sur moodle.

Notion de séquent

Les systèmes formels modélisant des logiques utilisent souvent un langage
basé sur une structure appelée séquent. Il s’agit d’une paire (Γ,F )
composée

d’un multi-ensemble de formules Γ (l’ordre ne compte pas, répétitions
possibles) et

d’une formule F .

Traditionnellement on note cette paire:

Γ ` F

Intuitivement, un séquent permet de représenter le fait que des hypothèses
Γ, on peut déduire F .

Notation Coq

En Coq au lieu d’écrire {A1,A2, . . . ,An} ` P on écrit:

H_1 : A_1

H_2 : A_2

H_n : A_n

___________________________________ (1/1)

P

Syntaxe

Logique Coq

⊥ False

> True

a = b a = b

a 6= b a <> b

¬A ~ A

A ∨ B A \/ B

A ∧ B A /\ B

A⇒ B A -> B

A⇔ B A <-> B

f (x , y , z) (f x y z)

∀xy ,P(x , y) forall (x y:A), P x y

∃xy ,P(x , y) exists (x:A) (y:B), P x y

Déduction naturelle

On utilise des séquents.

On ne manipule pas les hypothèses.

Règles pour la logique minimale

si A ∈ Γ
Γ ` A

Γ,A ` B
Intro →

Γ ` A→ B
Γ ` A→ B Γ ` A

Elim →
Γ ` B

Preuve de la formule K

A,B ` A
Intro →

A ` B → A
Intro →` A→ B → A

Preuve de la formule S

A→ B → C ,A→ B,A ` A→ B → C A→ B → C ,A→ B,A ` A
MP

A→ B → C ,A→ B,A ` B → C ... X ...
MP

A→ B → C ,A→ B,A ` C
Intro →

A→ B → C ,A→ B ` A→ C
Intro →

A→ B → C ` (A→ B)→ A→ C
Intro →` (A→ B → C)→ (A→ B)→ A→ C

X:

A→ B → C ,A→ B,A ` A→ B A→ B → C ,A→ B,A ` A
MP

A→ B → C ,A→ B,A ` B

Règles du ∧

Γ ` P Γ ` Q
Intro ∧

Γ ` P ∧ Q

Γ ` P ∧ Q
Elim ∧g

Γ ` P

Γ ` P ∧ Q
Elim ∧d

Γ ` Q

Règles du ∨

Γ ` P Intro ∨g
Γ ` P ∨ Q

Γ ` Q
Intro ∨d

Γ ` P ∨ Q

Γ ` P ∨ Q Γ,P ` R Γ,Q ` R
Elim ∨

Γ ` R

Règle du ¬

Γ,A ` ⊥
Intro ¬

Γ ` ¬A

Γ ` ¬A Γ ` A
Elim ¬

Γ ` ⊥

Remarque

On peut voir ¬A comme A→ ⊥.

Règles du ⊥

Γ ` ⊥
Elim ⊥

Γ ` P

Règles du ∀

Γ ` A ∀ intro (x n’est pas libre dans Γ)
Γ ` ∀x A

Γ ` ∀x A ∀ elim
Γ ` A[x ← t]

Règles du ∃

Γ ` A[x ← t]
∃ intro

Γ ` ∃x A

Γ ` ∃x A(x) Γ,A(x) ` B
∃ elim (x n’est pas libre dans Γ ni B)

Γ ` B

Convention de nommage

Remarque

Les règles sont nommées en les lisant de haut en bas.

Résumé
Γ,A ` B

Intro →
Γ ` A→ B

Γ ` P
Intro ∨g

Γ ` P ∨ Q

Γ ` P Γ ` Q
Intro ∧

Γ ` P ∧ Q

Γ,A ` ⊥
Intro ¬

Γ ` ¬A

Γ ` A→ B Γ ` A
Elim →

Γ ` B

Γ ` Q
Intro ∨d

Γ ` P ∨ Q

Γ ` P ∧ Q
Elim ∧g

Γ ` P

Γ ` ¬A Γ ` A
Elim ¬

Γ ` ⊥

si A ∈ Γ
Γ ` A

Γ ` P ∨ Q Γ,P ` R Γ,Q ` R
Elim ∨

Γ ` R

Γ ` P ∧ Q
Elim ∧d

Γ ` Q

Γ ` ⊥
Elim ⊥

Γ ` P

Γ ` ∃x A(x) Γ,A(x) ` B ∃elim
x 6∈ FV (Γ) ∪ FV (B)Γ ` B

Γ ` A ∀ intro (x 6∈ FV (Γ))
Γ ` ∀x A

Γ ` A[x ← t]
∃ intro

Γ ` ∃x A

Γ ` ∀x A ∀ elim
Γ ` A[x ← t]

Liens entre les tactiques Coq et les règles d’inférence

Etant donnée une règle d’inférence R comportant n prémisses de la forme:

Γ1 ` P1 . . . Γn ` Pn
R

Γ ` G

L’application de la tactique Coq correpondante consiste à transformer le
but:

Γ ` G

en n sous buts:
Γ1 ` P1

. . .

Γn ` Pn

Règle Axiome

si A ∈ Γ
Γ ` A

X : A

============

A

Proof completed.

assumption. ou apply X.

Règle Intro →
Γ,A ` B

Intro →
Γ ` A→ B

....

=====================

A -> B

X : A

=====================

B

intro X. ou intros. pour en faire plusieurs.

Règle Elim →
Γ ` A→ B Γ ` A

Elim →
Γ ` B

....

=====================

B

=====================

A -> B

=====================

A

cut A.

Variante Règle Elim →
Γ ` A Γ,A ` B

Γ ` B

....

=====================

B

...

=====================

A

...

A

=====================

B

assert (A).

Variante Elim →
Γ,H : A→ B ` A

Γ,H : A→ B ` B

H: A -> B

=====================

B

H: A -> B

=====================

A

apply H.

Variante Elim →
Γ,H : H1 → H2 → B ` A

Γ,H : A→ B ` B

H: H1 -> H2 -> B

=====================

B

H: H1 -> H2 -> B

=====================

H1

H: H1 -> H2 -> B

=====================

H2

apply H.

Règle Intro ∧
Γ ` A Γ ` B

Intro ∧
Γ ` A ∧ B

....

=====================

A /\ B

=====================

A

=====================

B

split.

Règle Elim ∧ g

Γ ` A ∧ B Elim ∧g
Γ ` A

....

=====================

A

=====================

A /\ B

assert (T: A /\ B);[idtac|elim T;intros;assumption].

Narboux et Magaud (Unistra) Preuves Assistées par Ordinateur 2018/2019 56 / 177



Règle Intro ∨ d

Γ ` B
Intro ∨d

Γ ` A ∨ B

....

=====================

A \/ B

=====================

B

right.

Règle Intro ∨ g

Γ ` A Intro ∨g
Γ ` A ∨ B

....

=====================

A \/ B

=====================

A

left.

Règle Elim ∨
Γ ` A ∨ B Γ,A ` G Γ,B ` G

Elim ∨
Γ ` G

Γ,H : A ∨ B,A ` G Γ,H : A ∨ B,B ` G

Γ,H : A ∨ B ` G

....

H : A \/ B

=====================

G

H : A \/ B

H0 : A

=====================

G

H : A \/ B

H0 : B

=====================

G

elim H;intro. ou bien destruct H ou bien decompose [or] H.

Variante Elim ∧
Γ,H : A ∧ B,H0 : A,H1 : B ` G

Γ,H : A ∧ B ` G

....

H : A /\ B

=====================

G

H : A /\ B

H0 : A

H1 : B

=====================

G

elim H;intro. ou bien destruct H ou bien decompose [and] H.

Règle Elim ⊥
Γ ` ⊥

Elim ⊥
Γ ` P

....

=====================

P

....

=====================

False

exfalso.

Règle Elim ⊥
Ax

Γ,H : ⊥ ` ⊥
Elim ⊥

Γ,H : ⊥ ` P

....

H : False

=====================

P

Proof completed.

elim H.

Règle Intro ¬
Γ,A ` ⊥

Intro ¬
Γ ` ¬A

....

=====================

~ A

H : A

=====================

False

intro.

Règle Elim ⊥
Γ ` ¬A Γ ` A

Elim ¬
Γ ` ⊥

Elim ⊥
Γ ` G

....

=====================

G

...

=====================

A

...

=====================

~ A

absurd A.

Quelques tactiques Coq

intro (introduire dans le context des hypothèses)

assert (prétendre qu’une proposition est vraie)

apply (applique un théorème)

exists (fournir un témoin pour prouver un ’exists’)

decompose [ex] H (exhibe les témoins des ’exists’ dans l’hypothèse
H)

decompose [and] H (coupe les ’et’ de H)

decompose [or] H (raisonne par cas sur les ’ou’ de H)

unfold t déplie une définition

simpl

reflexivity, symmetry, transitivity

rewrite, replace ... with

Composition de tactiques

tac0; tac1 applique la tactique tac0 sur le but courant, puis la
tactique tac1 sur les n sous-buts engendrés par la tactique tac0

Quelques notations

Logique Coq

∧ /\

∨ \/

¬ ~

⇒ ->

⇔ <->

∀ forall

∃ exists

Parenthèses: L’implique est parenthèsé à droite par défaut: (A→ B → C )
signifie (A→ (B → C )).

Utilisation de ∀
Quantifier sur les termes
forall n : nat, n >= 0

Quantifier sur les propositions

forall P Q : Prop, P \/ Q -> Q \/ P

Quantifier sur les prédicats

nat_ind: forall P : nat -> Prop,

P O ->

(forall n:nat, P n -> P (S n)) ->

forall n:nat, P n

Quantifier sur les types

eq_trans : forall (A : Type) (x y z : A), x = y /\ y = z -> x=z

fst : forall A B : Type, A * B -> A

Prop vs bool

Prop est le type des propositions que l’on prouve.
Exemples: True, False, 2<3, forall x, x=x,
leb 2 3 = true.

bool est le type des booléens utilisables dans des programmes
(dans des if).
Exemples: true, false, leb 2 3

Compute (leb 2 3).

= true

: bool

Compute (2 <= 3).

= 2 <= 3

: Prop

On peut prouver que: forall b: bool, b=true \/ b=false.
Mais on n’admet pas toujours que: forall P, P \/ ~P .

Fonctions sur les listes
Framework Map-Reduce

10 L’envers du décor

Logique classique vs intuitionniste

Les formules suivantes sont des formules valides en logique classique:

tiers exclu P ∨ ¬P
élimination de la double négation ¬¬P → P

loi de Peirce ((P → Q)→ P)→ P

Ces propositions impliquent qu’il y a des démonstrations qui ne
construisent pas l’objet satisfaisant la proposition prouvée.
Certains mathématiciens ont refusé ces propositions:

Brouwer,

Heyting, . . .

On dit qu’une preuve est constructive si elle n’utilise pas le tiers exclu.

Propriétés de la logique intuitioniste

Propriété de la disjonction

D’une preuve de A ∨ B on peut extraire une preuve de A ou une preuve de
B

Propriété du témoin

D’une preuve de ∃x ,A(x) on peut extraire un témoin t et une preuve de
A(t).

Exemple de preuve classique

Montrons que :

∃x , y 6∈ Q, xy ∈ Q

Considérons
√

2
√

2
.

a Si
√

2
√

2 ∈ Q.
On choisit x =

√
2 et y =

√
2.

b Sinon
√

2
√

2 6∈ Q.

On choisit x =
√

2
√

2
et y =

√
2.

xy = (
√

2

√
2
)

√
2

=
√

2

√
2×
√

2
=
√

2
2

= 2 ∈ Q

En fait, un théorème d’analyse affirme que
√

2
√

2
est irrationnel et que

c’est le cas “b“ qu’il faut choisir, mais la démonstration fondée sur le tiers
exclu ne le dit pas.

Logique propositionnelle classique

On rajoute la règle:

Réduction à l’absurde

Γ,¬P ` ⊥
RAA

Γ ` P

Remarque

On aurait pu écrire:

Γ ` ¬¬P
RAA’

Γ ` P

Exercice

Montrer que cette règle préserve la validité classique.

Exercices

1 ` ¬¬A→ A

2 ` A ∨ ¬A
3 ` ((A→ B)→ A)→ A

¬¬A,¬A ` ¬¬A ¬¬A,¬A ` ¬A
elim ¬¬¬A,¬A ` ⊥

RAA¬¬A ` A
intro →` ¬¬A→ A

¬(A ∨ ¬A) ` ¬(A ∨ ¬A)

¬(A ∨ ¬A), A ` A
intro ∨g

¬(A ∨ ¬A), A ` A ∨ ¬A ¬(A ∨ ¬A), A ` ¬(A ∨ ¬A)
elim ⊥

¬(A ∨ ¬A), A ` ⊥
intro ⊥

¬(A ∨ ¬A) ` ¬A
intro ∨d

¬(A ∨ ¬A) ` A ∨ ¬A

¬(A ∨ ¬A) ` ⊥
RAA

` A ∨ ¬A

(A → B) → A,¬A ` ¬A

(A → B) → A,¬A ` (A → B) → A

Γ,¬A, A ` A Γ,¬A, A ` ¬A

elim ¬
(A → B) → A,¬A, A ` ⊥

elim ⊥
(A → B) → A,¬A, A ` B

intro →
(A → B) → A,¬A ` A → B

elim →
(A → B) → A,¬A ` A

elim ⊥
(A → B) → A,¬A ` ⊥

RAA
(A → B) → A ` A

Intro →
` ((A → B) → A) → A

Coq est intuitionniste

Par défaut Coq travaille en logique intuitionniste.
Pour utiliser le tiers-exclu il faut rajouter l’axiome avec la commande:

Require Export Classical.

Exercice

Montrer que la formule suivante est valide / prouvable :

(A→ (B → C ))↔ (A ∧ B → C )

Remarque:
→ est associatif à droite, on aurait pu écrire:

(A→ B → C )↔ (A ∧ B → C )

Narboux et Magaud (Unistra) Preuves Assistées par Ordinateur 2018/2019 83 / 177



Curryfication

Définition

La curryfication consiste à transformer une fonction qui prend plusieurs
arguments en une fonction qui prend un seul argument et qui retourne une
fonction qui prend en arguments les arguments restants.

Remarque: cette opération porte le nom de Haskell Curry (1900-1982).

Narboux et Magaud (Unistra) Preuves Assistées par Ordinateur 2018/2019 84 / 177



Exemple

En Caml

Au lieu d’écrire:

# let f(x,y) = x + y;;

val f : int * int -> int = <fun>

On écrit:

# let f x = fun y -> x + y;;

val f : int -> int -> int = <fun>

ou bien

# let f x y = x + y;;

val f : int -> int -> int = <fun>

Narboux et Magaud (Unistra) Preuves Assistées par Ordinateur 2018/2019 85 / 177



Exemple

En Coq

On utilise la curryfication quand on écrit des fonctions ou des théorèmes.
On écrira plutôt:

Lemma toto: forall p q : R, p > 0 -> q > 0 -> p*q > 0.

que:

Lemma toto: forall p q : R, p > 0 /\ q > 0 -> p*q > 0.

Pourquoi ?

Pourquoi utiliser la curryfication ? Afin de pouvoir réaliser des applications
partielles plus facilement.

Sémantique de Heyting-Kolmogorov

La sémantique de Heyting-Kolmogorov consiste à donner une
interprétation fonctionnelle des démonstrations.

Une preuve de A→ B est une fonction qui, à partir d’une preuve de
A donne une preuve de B.

Une preuve de A ∧ B est une paire composée d’une preuve de A et
d’une preuve de B.

Une preuve de A ∨ B est une paire (i , p) avec (i = 0 et p une preuve
de A) ou (i = 1 et p une preuve de B).

Une preuve de ∀x .A est une fonction qui, pour chaque objet t
construit un objet de type A[x := t].

Cette interprétation consiste à calculer avec des preuves. Cela parâıt très
proche de la programmation fonctionnelle et du λ-calcul.

Exemple

Exemple du point de vue type:

Si H est de type A→ B et H ′ est de type A
alors

H H ′ est de type B.

Exemple du point de vue preuve:

Si H est une preuve de A→ B et H ′ est une preuve de A
alors

H H ′ est une preuve de B.

Fonctions sur les listes
Framework Map-Reduce

10 L’envers du décor

Correspondance de Curry-Howard I

logique programmation

Γ,A ` B
Γ ` A⇒ B

Γ, x : A ` t : B
Γ ` (fun x : A 7→ t) : A→ B

Γ ` A⇒ B Γ ` A
Γ ` B

Γ ` f : A→ B Γ ` a : A
Γ ` f (a) : B

Correspondance de Curry-Howard II

Γ ` A Γ ` B
Γ ` A ∧ B

Γ ` a : A Γ ` b : B
Γ ` (a, b) : A× B

Γ ` A ∧ B
Γ ` A

Γ ` t : A× B
Γ ` fst t : A

Γ ` A ∧ B
Γ ` B

Γ ` t : A× B
Γ ` snd t : B

Correspondance de Curry-Howard III

Γ ` A
Γ ` A ∨ B

Γ ` a : A
Γ ` inl a : A + B

Γ ` B
Γ ` A ∨ B

Γ ` b : B
Γ ` inr b : A + B

Γ ` A ∨ B Γ,A ` C Γ,B ` C

Γ ` C

Γ ` m : A ∨ B Γ, x : A ` t : C Γ, x : B ` u : C

Γ ` case m of inl(a) => t | inr(a) => u : C

Règles de simplification

fst(A,B) = A
snd(A,B) = B

case (inl m) of inl(a) => t|inr(a) => u = t[x := m]
case (inr m) of inl(a) => t|inr(a) => u = u[x := m]

Correspondance de Curry-Howard

Logique λ-calcul/programmation
formule type
preuve terme/programme

vérification d’une démonstration vérification de type
normalisation des preuves β-réduction

Extraction

Le mécanisme d’extraction transforme un terme Coq en un objet Caml.

Exemple:

Inductive list (A : Type) :=

nil : list A

| cons : A -> list A -> list A

Definition length (A : Type) :=

fix length l :=

match l with

| nil => O

| _ :: l’ => S (length l’)

end.

type ’a list =

| Nil

| Cons of ’a * ’a list

let rec length = function

| Nil -> O

| Cons (y, l’) ->

S (length l’)

Retour sur la correspondance de Curry-Howard à travers la
curryfication

Definition curry A B C (f:(A * B) -> C)

(x:A) (y:B) := f(x,y).

Lemma curry_prop: forall A B C : Type,

(A * B -> C) -> A -> B -> C.

Proof.

intros A B C f x y.

apply f.

split.

apply x.

apply y.

Defined.

Print curry_prop.

On obtient:

curry_prop =

fun (A B C : Type) (f : A * B -> C)

(x : A) (y : B) => f (x, y)

: forall A B C : Type,

(A * B -> C) -> A -> B -> C

Lemma uncurry_prop : forall A B C : Type,

(A -> B -> C) -> (A * B -> C).

Proof.

intros A B C f p.

elim p.

intros a b.

apply f.

apply a.

apply b.

Defined.

Extraction uncurry_prop.

On obtient:

let uncurry_prop f = function

| x,x0 -> f x x0

Exemple de construction d’un terme de preuve

Construire une preuve, sous forme d’un terme de la formule suivante:
A→ (A→ B)→ B.

En faire une formule close, à savoir ∀AB : Prop,A→ (A→ B)→ B.

On ne reste plus qu’à construire un terme du λ-calcul dont le type est
le suivant : ∀A B : Prop,A→ (A→ B)→ B.

I Ce sera une fonction avec pour arguments A, B, H1 et H2 et pour
corps un terme de type B construit à partir de A, B, H1 et H2.
fun (A:Prop) (B:Prop) (H1:A) (H2:A->B) => ...:B

I Une manière de construire un objet de type B est de prendre l’objet H1
de type A et de lui appliquer l’objet (de type fonctionnel) H2. Cela
donne le terme (applicatif) (H2 H1).

I Un terme de preuve possible pour ∀AB : Prop,A→ (A→ B)→ B est
donc fun (A:Prop) (B:Prop) (H1:A) (H2:A->B) => (H2 H1).

Exercices

On se donne:
and_ind : forall A B P : Prop,

(A -> B -> P) -> A /\ B -> P

conj : forall A B : Prop, A -> B -> A /\ B

or_ind : forall A B P : Prop,

(A -> P) -> (B -> P) -> A \/ B -> P

or_introl : forall A B : Prop, A -> A \/ B

or_intror : forall A B : Prop, B -> A \/ B

Construire le terme de preuve pour les types suivants:

1 (A→ B → C )→ (A→ B)→ A→ C

2 (A→ B)→ (B → C )→ (A→ C )

3 A ∧ B → B ∧ A

4 A ∧ B → A ∨ B

5 A ∨ B → B ∨ A

Types inductifs

Coq est basé sur un formalisme appelé Calcul des Constructions avec types
Inductifs.

Les définitions inductives consistent à:

se donner des éléments de base

et des règles pour construire de nouveaux éléments à partir
d’éléments déjà connus.

Premier exemple (sans récursion)

Inductive mois : Set :=

| Janvier : mois

| Fevrier : mois

| Mars : mois

| Avril : mois

| Mai : mois

| Juin : mois

| Juillet : mois

| Aout : mois

| Septembre : mois

| Octobre : mois

| Novembre : mois

| Decembre : mois.

Définition d’une fonction par filtrage

Definition mois_suivant (m:mois) : mois :=

match m with

| Janvier => Fevrier | Fevrier => Mars

| Mars => Avril | Avril => Mai

| Mai => Juin | Juin => Juillet

| Juillet => Aout | Aout => Septembre

| Septembre => Octobre | Octobre => Novembre

| Novembre => Decembre | Decembre => Janvier

end.

Comment calculer avec cette fonction ?
Eval compute in (mois suivant Mars).

Comment voir le code de cette fonction ?
Print mois suivant.

Définition du mois précédent

Definition mois_precedent (m:mois) : mois :=

match m with

| Janvier => Decembre | Fevrier => Janvier

| Mars => Fevrier | Avril => Mars

| Mai => Avril | Juin => Mai

| Juillet => Juin | Aout => Juillet

| Septembre => Aout | Octobre => Septembre

| Novembre => Octobre | Decembre => Novembre

end.

Premier lemme

Eval compute in (mois_precedent (mois_suivant Janvier)).

Lemma mois_prec_suiv : forall m:mois,

mois_precedent (mois_suivant m)=m.

Proof.

intros m.

elim m;

simpl;

reflexivity.

Qed.

elim

12 subgoals

m : mois

______________________________________(1/12)

mois_precedent (mois_suivant Janvier) = Janvier

______________________________________(2/12)

mois_precedent (mois_suivant Fevrier) = Fevrier

______________________________________(3/12)

mois_precedent (mois_suivant Mars) = Mars

...

______________________________________(12/12)

mois_precedent (mois_suivant Decembre) = Decembre

simpl

m : mois

______________________________________(1/12)

mois_precedent (mois_suivant Janvier) = Janvier

→ simpl →

m : mois

______________________________________

Janvier = Janvier

Raisonnement par cas

mois_ind : forall P : mois -> Prop,

P Janvier -> P Fevrier -> P Mars ->

P Avril -> P Mai -> P Juin ->

P Juillet -> P Aout -> P Septembre ->

P Octobre -> P Novembre -> P Decembre ->

forall m : mois, P m

Autre exemple:

Inductive bool : Type :=

| true : bool

| false : bool.

Aparté: l’égalité en Coq

Check eq.

eq : forall A : Type, A -> A -> Prop

Check refl_equal.

refl_equal : forall (A : Type) (x : A), x = x

C’est un type polymorphe (A:Type). Cela signifie que l’on a une
relation d’égalité générique dont le premier argument est le type des
éléments à comparer.

L’égalité est une relation réflexive, symétrique et transitive. Ces
propriétés sont utilisables grâce aux tactiques reflexivity,
symmetry et transitivity t.

Egalité de Leibniz

Check eq_ind.

eq_ind : forall (A : Type) (x : A) (P : A -> Prop),

P x -> forall y : A, x = y -> P y

Les tactiques pour l’égalité

rewrite

rewrite in

replace with

replace with in

subst

Aparté: Les sortes

Une sorte est un type pour les types.
Les propositions A, B, etc. sont des types (ceux de leurs termes de
preuves).
Ces types sont de type Prop.
On dit que A, B, etc. sont de sorte Prop.
D’un autre coté, les booléens bool, les entiers nat sont des types dont le
type est Set.
Set et Prop sont de type Type.

Set là où l’on calcule

Prop là où l’on raisonne

Les entiers de Peano

1 L’élément appelé zéro et noté: 0, est un entier naturel.

2 Si n est un entier naturel alors son successeur noté S(n) est un entier
naturel.

Autre notation: P ::= O | S P

Coq vs OCaml

Coq

Inductive nat : Set :=

O : nat

| S : nat -> nat.

Caml
type nat =

O

| S of nat

Récursion/Induction sur nat

forall P : nat -> Prop,

P 0 ->

(forall n : nat, P n -> P (S n)) ->

forall n : nat, P n

Quantification universelle sur une propriété P: nat -> Prop

Conclusion : forall n : nat, P n

2 cas:
1 cas de base
2 récurrence

forall n : nat, P n

________________

P 0

n : nat

IHn : P(n)

________________

P(S(n))

En Coq, un principe d’induction est automatiquement associé à chaque
définition inductive.

Autre exemple: les listes

Coq

Inductive list : Set :=

Nil : list

| Cons : nat -> list -> list.

Caml
type list =

Nil

| Cons of int*list

Principe d’induction

Check list ind.

list_ind

: forall P : list -> Prop,

P Nil ->

(forall (n : nat) (l : list), P l -> P (Cons n l)) ->

forall l : list, P l

Les constructeurs sont distincts

Pour cela, on dispose d’une tactique spécifique discrimate.

Lemma Mars_Fev : Mars<>Fevrier.

intro H; discriminate.

Qed.

Lemma zero_succ : forall n:nat, ~(S n)=O.

intros n H; discriminate H.

Qed.

Les constructeurs sont injectifs

Pour cela, on dispose d’une tactique spécifique injection.

Lemma test_injection: forall x y, S x = S y -> x=y.

Proof.

intros.

injection H.

intro.

assumption.

Qed.

Listes polymorphes

Inductive list (A:Set) : Set :=

nil : list A

| cons : A -> list A -> list A.

nil et cons ont un paramètre supplémentaire : A:Set.
Pour définir la liste [1;2] on obtient:

cons nat 1 (cons nat 2 (nil nat )).

Coq autorise les arguments implicites:

Implicit Arguments nil.

Implicit Arguments cons.

On peut alors écrire:

cons 1 (cons 2 nil ).

Et avec les notations:

1::2::nil.

Prédicats inductifs

Un prédicat inductif est une propriété définie inductivement.

Exemple: parité

Définition inductive
Inductive even : nat -> Prop :=

| even_O : even 0

| even_SS : forall m : nat, even m -> even (S (S m)).

Définition classique

Definition even’ n := exists k, n = 2*k.

Narboux et Magaud (Unistra) Preuves Assistées par Ordinateur 2018/2019 133 / 177



Connection

Lemma l1 : forall n, even n -> even’ n.

Définitions mutuellement récursives

Exemple

Inductive even : nat -> Prop :=

| even_O : even 0

| even_S : forall n, odd n -> even (S n)

with odd : nat -> Prop :=

odd_S : forall n, even n -> odd (S n).

Exemple: ≤

Définition inductive

Inductive le (n : nat) : nat -> Prop :=

le_n : le n n

| le_S : forall m : nat, le n m -> le n (S m).

Définition classique

Definition le (n m: nat) : Prop := exists p:nat, m=n+p.

Raisonnement par cas

inversion
Lemma toto : forall n,

le n 1 -> n = 1 \/ n =0.

Proof.

intros.

inversion H.

left.

reflexivity.

inversion H1.

right.

reflexivity.

Qed.

Deux cas:

2 subgoals

n : nat

H : le n 1

H0 : n = 1

______________________________________(1/2)

1 = 1 \/ 1 = 0

n : nat

H : le n 1

m : nat

H1 : le n 0

H0 : m = 0

______________________________________(2/2)

n = 1 \/ n = 0

inversion clear
Lemma toto : forall n,

le n 1 -> n = 1 \/ n =0.

Proof.

intros.

inversion_clear H.

left.

reflexivity.

inversion_clear H0.

right.

reflexivity.

Qed.

Conseils

Quand un argument peut être représenté sous forme d’un paramètre,
il vaut mieux le faire cela rend les inductions plus simples.

Inductive le (n : nat) : nat -> Prop :=

le_n : le n n

| le_S : forall m : nat, le n m -> le n (S m).

vs

Inductive le : nat -> nat -> Prop :=

le_n : le n n

| le_S : forall m : nat, le n m -> le n (S m).

Il est parfois utile d’avoir plusieurs versions d’une même définition
inductive, cela permet d’avoir plusieurs schémas de preuve possibles.

Tester vos prédicats sur des cas positifs et négatifs.

Exercice: clôture transitive-réflexive d’une relation R

Première Définition

a R∗ a
a R b
a R∗ b

a R∗ b b R∗ c
a R∗ c

Seconde Définition

a R# a
a R b b R# c

a R# c

1 Donner des prédicats inductifs correspondants à ces 2 définitions.

2 Montrer que si R est symétrique alors R∗ aussi.
3 On veut montrer que les deux définitions sont équivalentes.

1 Montrer que ∀a b, a R# b → a R∗ b.
2 Montrer que ∀a b, a R b → a R# b.
3 Montrer que R# est transitive.
4 En déduire que ∀a b, a R∗ b → a R# b.

Définition de fonctions

Maintenant qu’on a des structures de données on veut écrire des fonctions
qui les manipulent.

Caml Coq

let Definition
let rec Fixpoint

Fonction non récursive

Définition non récursive par filtrage:

Definition is_zero : nat -> bool :=

fun (n:nat) => match n with

O => true

| S _ => false

end.

ou bien

Definition is_zero (n:nat) :=

match n with

O => true

| S _ => false

end.

Fonction récursive: point fixe

Addition des entiers naturels:

Fixpoint plus (n m : nat) {struct n} : nat :=

match n with

| O => m

| S p => S (plus p m)

end.

Eval compute in (plus O 56).

Eval compute in (plus 12 67).

Eval compute in (plus 67 12).

Terminaison

En Coq les fonctions doivent toujours terminer !

L’annotation {struct n} indique l’argument qui décrôıt structurellement
à chaque appel récursif. Les appels récursifs se font sur des sous-termes
stricts. Cela permet de garantir la terminaison de la fonction.

Exemple

Fixpoint bidon (n:nat) : nat := bidon n.

Cette définition est refusée par le système et conduit à une erreur

Error: Recursive definition of bidon is ill-formed.

Sinon on aurait. . .

. . . une contradiction !

Fixpoint toto (b :bool) : bool := negb (toto b).

On a toto true = negb (toto true)

Si toto true = false alors toto true = true.

Si toto false = true alors toto false = false.

Totale

En Coq les fonctions doivent toujours être totales !

Solution

Sinon on peut utiliser le type option:

Inductive option (A:Type) : Type :=

| Some : A -> option A

| None : option A.

La profondeur du filtrage peut être supérieur à 1. On peut par exemple
définir une fonction pour tester la parité :

Fixpoint pair (n:nat) : Prop :=

match n with O => True

| (S O) => False

| (S (S p)) => pair p

end.

Eval compute in (pair 8789).

Eval compute in (pair 8790).

Exemple: Fibonacci

Fixpoint fib n {struct n} :=

match n with

O => 1

| S O => 1

| S ((S p) as p1) => fib p + fib p1

end.

Évaluation des fonctions

A la suite d’une définition par point-fixe un certain nombre de règles de
calcul (une par cas du filtrage) sont ajoutées dans le système.

tactiques compute, vm compute , simpl pour faire des réductions
dans le but courant.

simpl in H, pour faire la même chose dans une hypothèse plutôt
que pour le but courant.

Exemple:

Les réductions pour plus:
plus O m −→ι m

plus (S n) m −→ι S (plus n m)

Preuve de propriétés de programmes

Associativité de l’opération d’addition:

forall n m p : nat, (plus (plus n m) p) =

(plus n (plus m p))

Croissance de la fonction Fibonacci:

forall n : nat, fib n <= fib (S n)

Associativité de +

Démonstration par induction: intros n m p; elim n.

1 Cas de base : 0 + (m + p) = 0 + m + p
I étape de simplification : on utilise les règles de calcul pour les fonctions

mises en jeu (ici, uniquement +).
I conclusion de la démonstration : introductions, puis réflexivité de

l’égalité

2 Cas de récurrence.

Exercice

Définir une fonction d’itération telle que:

iter mois_suivant 12 m = m

Correction

Fixpoint iter (f:mois->mois) (n:nat) {struct n}

: mois -> mois :=

match n with

| O => (fun (m:mois) => m)

| (S p) => (fun (m:mois) => iter f p (f m))

end.

Lemma mois_12 : forall m : mois, iter mois_suivant 12 m = m.

intros m; case m; simpl; trivial.

Qed.

Narboux et Magaud (Unistra) Preuves Assistées par Ordinateur 2018/2019 155 / 177



Terminaison

Merge sort

Fixpoint merge (l1:list nat) (l2: list nat) :=

match (l1,l2) with

| (nil,_) => l2

| (_,nil) => l1

|(a::m1,b::m2) => if (le_lt_dec a b) then a::merge m1 l2

else b::merge l1 m2

end.

Cette définition n’est pas acceptée.

Error: Cannot guess decreasing argument of fix.

Première solution

On utilise un compteur:

Fixpoint merge (l1:list nat) (l2: list nat) (n:nat):=

match (l1,l2,n) with

| (nil,_,_) => l2

| (_,nil,_) => l1

|(_,_,0) => nil

|(a::m1,b::m2,S p) => if (le_lt_dec a b) then a::merge m1 l2 p

else b::merge l1 m2 p

end.

Inconvenients

Il faut calculer la longueur des listes.

Les preuves utilisant la fonction doivent prouver que le compteur a
bien été initialisé.

Deuxième solution

On utilise Function:
Require Export Recdef.

Definition length_sum (c:list nat * list nat) :=

length (fst c) + length (snd c).

Function merge’ (c:list nat*list nat)

{measure length_sum} : list nat :=

match c with

| (nil,l2) => l2

| (l1,nil) => l1

|(a::m1 as l1,b::m2 as l2) => if (le_lt_dec a b)

then a::merge’ (m1,l2)

else b::merge’ (l1,m2)

end.

Obligations de preuve

Il faut prouver que:

length_sum (l0, l3) < length_sum (a :: l0, b :: l3)

La preuve:

Proof.

intros.

unfold length_sum.

simpl.

omega.

intros.

unfold length_sum.

simpl.

omega.

Qed.

Fonction vs prédicat

On peut utiliser un prédicat pour modéliser une fonction (comme en
Prolog):

Exemple: Siracuse

U0 = N Un+1 =

{
Un
2 si Un est pair

3Un + 1 si Un est impair

Inductive syracuse (N:nat) : nat -> nat -> Prop :=

done : syracuse N 0 N

| even_case : forall n p,even p ->

syracuse N n p -> syracuse N (S n) (div2 p)

| odd_case : forall n p , odd p ->

syracuse N n p -> syracuse N (S n) (S(p+p+p)).

Siracuse :utilisation

Lemma essai: syracuse 15 1 46.

replace (46) with (S(15+15+15)) by reflexivity.

apply odd_case.

repeat constructor.

constructor.

Qed.

Quelques fonctions incontournables sur les listes

1 filter

2 map

3 fold/reduce

Filter

La fonction filter prend en argument un prédicat et une liste et renvoie
la liste des éléments de la liste qui vérifie le prédicat.

Fixpoint filter {X:Type}(test: X->bool)(l:list X)

:(list X) :=

match l with

| [] => []

| h :: t => if test h then h :: (filter test t)

else filter test t

end.

Exemple

Eval simpl in (filter (fun x => leb x 3) (1::20::3::nil)).

= 1 :: 3 :: nil : list nat

Map

La fonction map applique une fonction à chaque élément d’une liste:

Fixpoint map {X Y:Type} (f:X->Y) (l:list X):(list Y):=

match l with

| [] => []

| h :: t => (f h) :: (map f t)

end.

Exemple

Eval simpl in (map (plus 2) (1::2::3::4::nil)).

= 3 :: 4 :: 5 :: 6 :: nil : list nat

Fold

Intuitivement, la fonction fold consiste à insérer un opérateur binaire
entre chaque paire d’éléments consécutifs d’une liste.

Fixpoint foldr {X Y:Type} (f: X->Y->Y) (l:list X) (b:Y):Y :=

match l with

| nil => b

| h :: t => f h (foldr f t b)

end.

Fold

Exemple

Eval simpl in (fold plus (1::2::3::4::nil) 0).

= 10 : nat

Aparté: Framework Map-Reduce

Ces fonctions ont inspirées le framework map/reduce de Google ou
Hadoop de la fondation Apache:

”Our abstraction is inspired by the map and reduce primitives
present in Lisp and many other functional languages. We realized
that most of our computations involved applying a map
operation to each logical record in our input in order to compute
a set of intermediate key/value pairs, and then applying a reduce
operation to all the values that shared the same key in order to
combine the derived data appropriately.”

Jeffrey Dean and Sanjay Ghemawat

Exemple I

Compter le nombre d’occurences d’une châıne de caractère s dans un
ensemble de documents:

Definition sum l := fold_left plus l 0.

Definition compte_occs s docs :=

sum (map (compte_occ s) docs).

Règles de la déduction naturelle
Liens entre les tactiques Coq et la déduction naturelle

Narboux et Magaud (Unistra) Preuves Assistées par Ordinateur 2018/2019 169 / 177



En Coq, certains connecteurs logiques sont primitifs et d’autres sont
définis par des types inductifs:

Définis:

>
⊥
∧
∨
∃
¬A ≡ A⇒ ⊥

Primitifs:

∀
⇒

Vrai

Inductive True : Prop :=

I : True.

True_ind

: forall P : Prop, P -> True -> P

Inutile. . .

Faux

Inductive False : Prop := .

False_ind

: forall P : Prop, False -> P

Exemple

Theorem ex_falso_quodlibet : forall (P:Prop),

False -> P.

Proof.

intros P F.

inversion F.

Qed.

Et

Inductive et (A: Prop) (B: Prop) : Prop :=

| Conj : A -> B -> et A B.

et_ind

: forall A B P : Prop,

(A -> B -> P) -> et A B -> P

Ou

Inductive ou (A: Prop) (B : Prop) : Prop :=

| ouintrog : A -> ou A B

| ouintrod : B -> ou A B.

ou_ind

: forall A B P : Prop,

(A -> P) -> (B -> P) -> ou A B -> P

Le quantificateur existentiel

Inductive ex (A: Set) (P: A -> Prop) : Prop :=

| ex_intro : forall x : A, P x -> ex A P.

ex_ind

: forall (A : Set) (P : A -> Prop) (P0 : Prop),

(forall x : A, P x -> P0) -> ex A P -> P0

Égalité

Inductive eq (A:Type) (x:A) : A -> Prop :=

refl_equal : eq A x x.

eq_ind: forall (A:Type)(x:A)(P:A->Prop),

P x -> forall y: A, x=y -> P y

